

Занятие 8 — Система символьных вычислений Maple

Программирование, численные методы и информатика

А. В. Позднеев

Кафедра автоматизации научных исследований
Факультет вычислительной математики и кибернетики
Московский государственный университет имени М. В. Ломоносова
<http://ani.smc.msu.ru/geol>

Осенний семестр 2009/2010 уч. г.



Содержание занятия

- 1 Maple - общие сведения
- 2 Основы работы в Maple
- 3 Преобразование выражений
- 4 Дифференцирование
- 5 Интегрирование
- 6 Построение графиков
- 7 Решение уравнений и неравенств
- 8 Дифференциальные уравнения
- 9 Интерполирование функций
- 10 Вычисление пределов
- 11 Суммы и произведения
- 12 Упражнения

Maple — общие сведения

Общая информация:

- ▶ Maple — система компьютерной алгебры
- ▶ Символьные вычисления (преобразование выражений с использованием переменных, многочленов, функций, ...)
- ▶ Продукт компании Maplesoft (<http://www.maplesoft.com>)
- ▶ Аналоги: Mathematica (коммерческая), Axiom (свободная), Maxima (свободная), Sage (свободная)

Актуальная для нас информация:

- ▶ Текущая версия — 13.02 (октябрь 2009)
- ▶ В курсе рассматривается версия 10.00
- ▶ Используется вариант *Classic Worksheet Maple*
- ▶ В примерах в данных слайдах использованы примеры из лекций доц. В. В. Нефёдова и встроенной справочной системы Maple

Справочная система

- ▶ Навести курсор на команду и нажать F1
- ▶ Ввести команду

```
> ?command
```

Например:

```
> ?int
```

Меню «Help»:

- ▶ Help → Topic Search — поиск по названию темы
- ▶ Help → Full Text Search — полнотекстовый поиск

Ввод команд

- ▶ Работа в режиме сессии — команды интерпретируются сразу же
- ▶ Команды вводятся после символа «больше» ($>$)
- ▶ Если команда завершается точкой с запятой ($;$), то после нажатия `Enter` результат сразу же выводится на экран
- ▶ Если — двоеточием ($:$), то команда выполняется, но результат не выводится на экран

```
a := 1:  
b := a + 2;
```

- ▶ Итак, нажатие `Enter` приводит к выполнению команды
- ▶ Текущую сессию можно сохранить для того, чтобы продолжить с ней работу в другой раз (`Ctrl + S` или меню `File → Save`)

«Горячие клавиши»

Основные

- ▶ Shift + Enter — перейти на следующую строчку, не выполняя текущую команду
- ▶ Ctrl + F4 — закрыть текущее окно
- ▶ Ctrl + F6, Ctrl + Tab — перейти к следующему окну
- ▶ F6, Ctrl + пробел — автодополнение частично введенной команды

Дополнительные

- ▶ Ctrl + . — заключить выделенные команды в «сворачиваемую» секцию
- ▶ Ctrl + N — открыть новую сессию

Полный список «горячих клавиш» — в разделе помощи
«`shortcut_keys, windows`»

Простейшие выражения

```
2 * 2;           # обычные арифметические операции
5!;             # факториал
2^3;           # возведение в степень
1/2 - 1/3;      # операции с обыкновенными дробями
evalf(%);       # приближенные вычисления
evalf(%%, 30);  # любое число знаков после запятой
1.0/2 - 1/3;    # десятичные дроби => приближенно
```

Можно ссылаться на результаты предыдущих вычислений

- ▶ % — последнее вычисленное выражение
- ▶ %% — предпоследнее вычисленное выражение
- ▶ %%% — выражение, вычисленное перед предпоследним выражением
- ▶ обозначения %%% не существует
- ▶ restart — очистить внутреннюю память

Предопределенные константы

- ▶ `Pi` — число π
- ▶ `I` — число i ($i^2 = -1$)
- ▶ `infinity` — бесконечность (∞)
- ▶ `true` — логическое значение «истина»
- ▶ `false` — логическое значение «ложь»

```
sin(Pi/6);  
sin(evalf(Pi/6));  
exp(I*Pi/6); # cos(Pi/6) + I*sin(Pi/6);  
limit((1+1/x)^x, x=infinity);  
limit(exp(x), x=-infinity);
```


Операторы и переменные

Операторы

- ▶ Арифметические: +, -, *, /, ^, **, !
- ▶ Логические: <, <=, >, >=, =, <>
- ▶ Оператор присваивания: :=,

Переменные

- ▶ Переменная — любой идентификатор, состоящий из латинских букв и цифр, начинающийся с буквы и не совпадающий ни с одним из зарезервированных слов
- ▶ Переменной может быть присвоено любое значение при помощи оператора присваивания :=
- ▶ Переменная, которой не присвоено никакое значение, считается свободной переменной и ее имя сохраняется в арифметических вычислениях

```
a := 2;  
b := 3;  
(a+b)^2;  
(a+c)^2;
```

Стандартные функции

- ▶ $\text{sqrt}(x)$ — квадратный корень
- ▶ $\text{abs}(x)$ — абсолютная величина (модуль)
- ▶ $\text{sign}(x)$, $\text{signum}(x)$, $\text{csgn}(x)$ — знак выражения
- ▶ $\text{sin}(x)$, $\text{cos}(x)$, $\text{tan}(x)$, $\text{cot}(x)$ — тригонометрические функции
- ▶ $\text{sinh}(x)$, $\text{cosh}(x)$, $\text{tanh}(x)$, $\text{coth}(x)$ — гиперболические функции
- ▶ $\text{arcsin}(x)$, $\text{arccos}(x)$, $\text{arctan}(x)$, $\text{arccot}(x)$ — обратные тригонометрические функции
- ▶ $\text{exp}(x)$ — экспонента
- ▶ $\text{ln}(x)$, $\text{log}(x)$ — натуральный логарифм
- ▶ $\text{log}[b](x)$ — логарифм по основанию b
- ▶ $\text{log}_{10}(x)$ — десятичный логарифм

```
y := log[a](x);
```

Преобразование выражений. Факторизация

- ▶ `ifactor(n)` — разложение рационального числа на множители

```
ifactor(13324104179219423947014392919);  
ifactor(34324131214142 / 123411112432413211);
```

- ▶ `factor(a)` — разложение выражения a на множители

```
factor(x^2 - 3*x + 2);  
P := x^4 + 2*x^3 + 2*x^2 + 2*x + 1;  
factor(P);
```

- ▶ `factor(a, K)` — разложение «над полем K »

```
factor(y^4-2);  
factor(y^4-2, sqrt(2));  
factor(y^4-2, sqrt(sqrt(2)));  
factor(y^4-2, {sqrt(sqrt(2)), sqrt(-sqrt(2))});  
factor(y^4-2, real);  
factor(y^4-2, complex);
```

Преобразование выражений. Раскрытие скобок

- ▶ `expand(expr)` — раскрыть скобки в выражении `expr`

```
expand((x+1)*(x+2));  
expand((x+1)/(x+2));  
expand(sin(x+y));  
expand(cos(2*x));  
expand(exp(a+ln(b)));
```

- ▶ `expand(expr, expr1)` — НЕ раскрывать скобки в `expr1`

```
expand((x+1)*(y+z));  
expand((x+1)*(y+z), x+1);
```

Преобразование выражений. Нормализация

Приведение дробей к общему знаменателю с последующим сокращением

- ▶ `normal(f)` — нормализовать дробь f

```
(a^4-b^4)/((a^2+b^2)*a*b);  
normal(%);  
normal(1/x + 1/y);  
normal(1/x + x/(x+1));  
normal((x^2-y^2)/(x-y)^3);
```

- ▶ `normal(f, expanded)` — нормализовать и раскрыть скобки

```
normal(1/x+x/(x+1), expanded);  
normal((x^2-y^2)/(x-y)^3, expanded);
```

Преобразование выражений. Упрощение выражений

- ▶ `simplify(expr)` — упростить выражение `expr`

```
simplify(4^(1/2)+3);  
simplify(exp(a+ln(b*exp(c))));  
simplify(sin(x)^2+ln(2*x)+cos(x)^2);
```

- ▶ `simplify(expr, n1)` — упростить, используя правило `n1`

```
simplify(sin(x)^2+ln(2*x)+cos(x)^2, trig);  
simplify(sin(x)^2+ln(2*x)+cos(x)^2, ln);
```

- ▶ `simplify(expr, assume=prop)` — упростить при условии `prop`

```
g := sqrt(x^2);  
simplify(g);  
simplify(g, assume=real);  
simplify(g, assume=positive);
```

Преобразование выражений. Прочие функции

- ▶ `collect()` — собрать коэффициенты при степенях
- ▶ `combine()` — собрать «подобные» слагаемые

Дифференцирование выражений

- ▶ `diff(f, x)` — найти производную выражения по переменной x

```
diff(sin(x), x);  
p := a * x^2;  
diff(p, x);  
diff(p, a);
```

- ▶ `diff(f, x$n)` — продифференцировать n раз по x

```
diff(cos(x), x$2);  
diff(sin(x), x$n);  
diff(exp(a*x), x$n);
```

- ▶ `diff(f, x1, ..., xn)` — продифференцировать по x_1, \dots, x_n

```
diff(exp(x)*sin(y), x, y, y);
```


Первообразная и определенный интеграл

- ▶ `int(f, x)` — неопределенный интеграл от f по dx

```
int(sin(x), x);  
int(a * x^2, x);  
int(a * x^2, a);
```

- ▶ `int(f, x=a..b)` — определенный интеграл $\int_a^b f dx$

```
int(sin(x), x=0..Pi);  
int(sin, a..b);  
int(exp(-x^2), x=0..infinity);
```

- ▶ Для приближенного вычисления воспользуйтесь `evalf`

```
g := int(tan(x^2), x=1/3..1/2);  
evalf(g, 3);
```

Интегралы, зависящие от параметра

```
assume(a > -1);           # задать ограничение на параметр
additionally(a <= 3);     # указать дополнительное ограничение
about(a);                 # вывести информацию об ограничении
```

```
restart;
int(exp(-a*x), x=0..+infinity);
assume(a > 0);
int(exp(-a*x), x=0..+infinity);
```

Обучение интегрированию

- ▶ `intparts(f, u)` — проинтегрировать по частям выражение f вида `Int(u*dv, x)`. В качестве результата возвращает $u*v - \text{Int}(v*du, x)$

$$\int u dv = uv - \int v du$$

```
with(student):  
g := Int(x^2 * cos(x), x);  
intparts(g, x^2);  
intparts(%, x);  
value(%)
```

- ▶ `changevar(s, f, u)` — проинтегрировать по частям выражение f вида `Int(F(x), x)`

```
with(student):  
g := Int(1/(sqrt(x) * (1+x)), x);  
changevar(sqrt(x) = t, g, t);  
value(%)
```

Демонстрация численных методов интегрирования

Графическое изображение численного метода интегрирования

```
restart;  
with(Student[Calculus1]):  
g := 1/x:  
a := 1:  
b := 4:  
ApproximateInt(g, x=a..b, method=left, output=plot);  
ApproximateInt(g, x=a..b, method=right, output=plot);  
ApproximateInt(g, x=a..b, method=midpoint, output=plot);  
ApproximateInt(g, x=a..b, method=trapezoid, output=plot);  
ApproximateInt(g, x=a..b, method=simpson, output=plot);
```

Интерактивная форма

```
restart;  
with(Student[Calculus1]):  
ApproximateIntTutor();
```

Построение графиков

- ▶ Функция одной переменной

```
plot(sin(x), x=-Pi..Pi);  
plot(1/x, x=-3..3, -10..10);  
plot([x^2, x^3], x=-2..2);
```

- ▶ Функция, заданная параметрически

```
plot([cos(t), sin(t), t=0..3/2*Pi]);
```

- ▶ График в полярных координатах

```
r := 1:  
phi := t:  
plot([r, phi, t=0..3/2*Pi], coords=polar);
```

- ▶ Функция двух переменных

```
plot3d(x*sin(y), x=-2..2, y=-Pi..Pi);
```

Решение уравнений в целых числах

- ▶ `isolve(e)` — решить диофантово уравнение

```
isolve(x+y=0);
```

- ▶ Уравнение Пелля $x^2 - ny^2 = 1$ при $n = 2$

```
restart;  
sol := isolve(x^2 - 2*y^2 = 1);  
_Z1:=0: sol;  
_Z1:=1: sol;  
_Z1:=2: sol;  
expand(sol);
```

- ▶ Пифагоровы тройки

```
unassign('_Z1');  
sol := isolve(x^2 + y^2 = z^2);  
_Z1:=1: _Z2:=3: _Z3:=1: sol;
```

Алгебраические и тригонометрические уравнения

- ▶ `solve(f, x)` — решить уравнение f относительно переменной x

```
solve(x^2 - 3*x + 2, x);  
solve(a*x + b = 0, x);  
solve(a*x + b = 0, b);  
p := solve(x^2 + x - 1 = 0, x); p[1]; p[2];  
p[1]*p[2];  
simplify(p[1]*p[2]);
```

- ▶ `solve({f1, f2}, {x1, x2})` — решить систему уравнений

```
p := solve({x+y = a, x-y = b}, {x, y});  
assign(p);  
x;  
y;
```

- ▶ Решение тригонометрических уравнений

```
_EnvAllSolutions := true;  
solve(sin(2*x) + cos(2*x) = 1, x);
```

Решение неравенств

- ▶ `solve({f1, f2}, {x1, x2})` — решить систему неравенств

```
solve({x + y < 10, x^2 = 9}, {x, y});
```

- ▶ Графический метод решения системы линейных неравенств

```
with(Student[Precalculus]):  
LinearInequalitiesTutor();
```


Численное решение уравнений

- ▶ `fsolve(f)` — решить уравнение `f`

```
f := x^6 - 2*x + 1 = 0;  
solve(f, x);  
fsolve(f, x);  
fsolve(exp(-x) = x);  
fsolve(cos(x) = x);
```

Решение обыкновенных дифференциальных уравнений

$$\begin{cases} \frac{d^2}{dx^2}y(x) = y(x) + 1, \\ y(0) = 1, \\ \frac{d}{dx}y(0) = 0. \end{cases}$$

- ▶ `dsolve(ode, y(x))` — общее решение ОДУ `ode` относительно `y(x)`

```
ode := diff(y(x), x, x) = y(x) + 1;  
dsolve(ode, y(x));
```

- ▶ `dsolve(ode, ics, y(x))` — решение задачи Коши с начальными значениями `ics`

```
ics := y(0)=1, D(y)(0)=0;  
dsolve({ode, ics}, y(x));
```

Решение краевых задач для ОДУ


$$\begin{cases} \frac{d^2}{dx^2}y(x) + 3\frac{d}{dx}y(x) + 2y(x) = 0, \\ y(0) = 0, \\ y(1) = 1. \end{cases}$$

```
ode := diff(y(x), x$2) + 3*diff(y(x), x) + 2 * y(x);  
bcs := y(0) = 0, y(1) = 1;  
dsolve({ode, bcs}, y(x));
```

Численное решение ОДУ


$$\begin{cases} \frac{d}{dx}y(x) = y^2(x) + x^2, \\ y(1) = 1, \end{cases}$$

```
ode := diff(y(x), x) = y(x)^2 + x^2;  
dsolve(ode);  
ics := y(1) = 1;  
sol := dsolve({ode, ics}, y(x), numeric);  
with(plots):  
plots[odeplot](sol, [x, y(x)], 1..1.7, labels=[x, y]);
```




Интерполирование функций. Интерполяционный многочлен Лагранжа

```
restart;  
with(CurveFitting):  
x := [0, 1, 2, 3, 4, 5]:  
y := [0, 1, 4, 3, 2, 1]:  
L := PolynomialInterpolation(x, y, t, form=Lagrange);  
plot(L, t=0..5);
```



Интерполирование функций. Сплайн-интерполяция

```
restart;  
with(CurveFitting):  
x := [0, 1, 2, 3, 4, 5]:  
y := [0, 1, 4, 3, 2.5, 10]:  
s1 := Spline(x, y, t, degree=1);  
s2 := Spline(x, y, t, degree=2);  
s3 := Spline(x, y, t, degree=3);  
s4 := Spline(x, y, t, degree=4);  
plot([s1, s2, s3, s4], t=-1..6);
```



Интерполирование функций. Метод наименьших квадратов

```
restart:
with(stats):
Xvalues := [0, 1, 2, 3, 4, 5]:
Yvalues := [0, 1, 4, 3, 2.5, 10];
ls := fit[leastsquare][[x, y], y=a*x+b]([Xvalues, Yvalues]);
lsproc := unapply(rhs(ls), x);
plot(lsproc(t), t=-1..6);
```


Вычисление пределов

- ▶ $\text{limit}(f, x=a)$ — вычислить предел выражения f при $x \rightarrow a$

```
limit(sin(x)/x, x=0);  
limit((1 + x)^(a/x), x=0);
```

- ▶ Односторонние пределы

```
limit(abs(x)/x, x=0, left);  
limit(abs(x)/x, x=0, right);
```



Суммы и произведения

- ▶ `sum(f(k), k=a..b)` — сумма

$$\sum_{k=a}^b f(k) = f(a) + f(a+1) + \dots + f(b)$$

```
sum(k, k=1..100);  
sum(k^2, k=1..100);  
sum(q^k, k=0..infinity);  
sum(1/k^2, k=1..infinity);
```

- ▶ `product(f(k), k = m..n)` — произведение

$$\prod_{k=a}^b f(k) = f(a) \times f(a+1) \times \dots \times f(b)$$

```
product(i, i=1..6);  
product(1+1/k^2, k=1..infinity);
```

Упражнения

Простейшие выражения. Преобразование выражений. Решение уравнений

1. Вычислить значение $(6 + 2 \cdot 5^{1/2})^{1/2} - (6 - 2 \cdot 5^{1/2})^{1/2}$
2. Вычислить $\sin^4 \frac{\pi}{8} + \cos^4 \frac{3\pi}{8} + \sin^4 \frac{5\pi}{8} + \cos^4 \frac{7\pi}{8}$
3. Упростить выражение $\frac{1 + \sin(2x) + \cos(2x)}{1 + \sin(2x) - \cos(2x)}$
4. Разложить на множители многочлен $x^3 - 4x^2 + 5x - 2$
5. Найти численное решение уравнения $\cos x = x$
6. Решить уравнение $3x - (18x + 1)^{1/2} + 1 = 0$
7. Решить уравнение $||2x - 3| - 1| = x$
8. Найти все решения уравнение $\sin x - \cos x = \frac{1}{\sin x}$
9. Решить систему алгебраических уравнений

$$\begin{cases} 10(xy)^{1/2} + 3x - 3y = 58 \\ x - y = 6 \end{cases}$$

Упражнения

Построение графиков. Нахождение пределов. Факторизация. Решение уравнений в целых числах

1. Построить график функции $y = x \sin \frac{1}{x}$ на сегменте $[\varepsilon, a]$.
В качестве параметра ε выбрать несколько значений (например, $\varepsilon = 0,001, 0,01, 0,1$ и т. п.). В качестве параметра a взять значения 1, 2, 5, 10. Будет ли построен график заданной функции при $\varepsilon < 0$? Вычислить предел функции при $x \rightarrow 0$.
2. Потренироваться в построении графиков различных элементарных функций, известных из курса математического анализа.
Попытаться изобразить на одном графике несколько различных функций (использовать для их изображения линии разных типов).
3. Разложить на множители число $10^{10} + 1$.
4. Найти общий вид решения в целых числах уравнения $3x + 5y = 179$.
5. Найти общий вид решения в натуральных числах уравнения $x^2 - 2y^2 = -1$. Найти первых три наименьших его решения.

Упражнения

Интегрирование

1. Вычислить неопределенный интеграл $\int \frac{x^3-6}{x^4+6x^2+8} dx$
2. Вычислить несобственный интеграл $\int_0^{\infty} \frac{\sin ax \cos bx}{x} dx$ при $a > 0$, $b > 0$ для трех случаев: $a < b$, $a = b$, $a > b$
3. Найти численно (т. е. приближенно) интеграл $\int_{0,1}^{0,2} \frac{\sin 3xe^{-x^2}}{x^4} dx$
4. Полностью проделать все этапы вычисления определенного интеграла $\int_0^{\pi/2} x^3 \cos x dx$, используя для этого формулу интегрирования по частям
5. Найти значение интеграла $\int_0^{\pi/2} \frac{dx}{5-4 \sin x+3 \cos x}$, используя для этого универсальную тригонометрическую подстановку $\operatorname{tg} \frac{x}{2} = t$